

# **CATC Scripting Language Reference Manual for CATC Bluetooth Analyzers**

**Manual Ver. 1.21**

10 February, 2004

# **CATC Scripting Language Reference Manual, Version 1.21**

## **Document Disclaimer**

The information contained in this document has been carefully checked and is believed to be reliable. However, no responsibility can be assumed for inaccuracies that may not have been detected.

CATC reserves the right to revise the information presented in this document without notice or penalty.

## **Trademarks and Servicemarks**

CATC is a trademark of Computer Access Technology Corporation.

All other trademarks are property of their respective companies.

## **Copyright**

Copyright 2004, Computer Access Technology Corporation (CATC). All rights reserved.

This document may be printed and reproduced without additional permission, but all copies should contain this copyright notice.

# TABLE OF CONTENTS

<b>1</b>	<b>Introduction</b> .....	<b>1</b>
	Support for CSL in Bluetooth Products .....	1
	Contents of this manual .....	2
<b>2</b>	<b>Values</b> .....	<b>3</b>
	Literals .....	3
	Integers .....	3
	Strings .....	3
	Escape Sequences .....	4
	Lists .....	4
	Raw Bytes .....	4
	Null .....	4
	Variables .....	5
	Global Variables .....	5
	Local Variables .....	6
	Constants .....	6
<b>3</b>	<b>Expressions</b> .....	<b>7</b>
	select expression .....	7
<b>4</b>	<b>Operators</b> .....	<b>9</b>
	Operations .....	9
	Operator Precedence and Associativity .....	9
<b>5</b>	<b>Comments</b> .....	<b>15</b>
<b>6</b>	<b>Keywords</b> .....	<b>17</b>
<b>7</b>	<b>Statements</b> .....	<b>19</b>
	Expression Statements .....	19
	if Statements .....	19
	if-else Statements .....	19
	while Statements .....	20
	for Statements .....	20
	return Statements .....	21
	Compound Statements .....	22

<b>8 Preprocessing</b> .....	<b>25</b>
<b>9 Context</b> .....	<b>27</b>
<b>10 Functions</b> .....	<b>29</b>
<b>11 Primitives</b> .....	<b>31</b>
Call() .....	31
Format() .....	31
Format Conversion Characters .....	32
GetNBits() .....	33
GetBitOffset() .....	34
PeekNBits() .....	34
NextNBits() .....	35
Resolve() .....	36
Trace() .....	36
<b>12 Decoder Primitives</b> .....	<b>39</b>
Abort() .....	39
AddCell() .....	39
AddDataCell() .....	41
AddEvent() .....	42
AddSeparator() .....	43
BeginCellBlock() .....	43
Complete() .....	45
EndCellBlock() .....	46
Pending() .....	46
Reject() .....	47
<b>13 Modules</b> .....	<b>49</b>
Module Functions .....	49
ProcessData() .....	49
CollectData() .....	49
BuildCellList() .....	49
Module Data .....	50
ModuleType .....	50
OutputType .....	50
InputType .....	50
LevelName .....	50
DecoderDesc .....	50
Icon .....	51

LelelColor .....	51
<b>Appendix A: Bluetooth Data .....</b>	<b>53</b>
<b>Appendix B: Sample BNEP Script .....</b>	<b>55</b>



# CHAPTER 1: INTRODUCTION

CATC Scripting Language (CSL) allows users to create scripts that perform file-based decoding with CATC analyzers. CSL is used to edit CATC Decode Scripting (CDS) files. CATC analyzers are equipped with built-in decoders in addition to script-based decoders that can be modified by the users or implemented as-is. Decoding scripts for analyzers are distributed in the `/scripts` directory. These scripts are tools to decode and display transactions. Users can also add entirely new, customized decoders.

CSL is based on C language syntax, so anyone with a C programming background will have no trouble learning CSL. The simple, yet powerful, structure of CSL also enables less experienced users to easily acquire the basic knowledge needed to start writing custom scripts.

## *Features of CATC Scripting Language*

- A powerful, high-level API that allows implementation of complex algorithms.
- Easy to learn and use -- has a simple but effective syntax.
- Self-contained -- needs no external tools to run scripts.
- Wide range of value types -- provides efficient and easy processing of data.
- Used to create built-in script-based decoders for analyzers.
- May be used to write custom decoders.
- General purpose -- is integrated in a number of CATC products.

## **Support for CSL in Bluetooth Products**

The scripting language described in this document is supported by all CATC Bluetooth protocol analyzers:

### **BTTracer software 2.0**

### **Merlin software 1.50**

### **Merlin Mobile software 1.0**

As the CATC Scripting language evolves over time to supports more functionalities and deliver even greater flexibility, older releases of the analyzers software's do not support all the latest upgrades. For newer releases of the analyzer's software, please refer to CATC's web site:

**[www.catc.com/support](http://www.catc.com/support)**

For additional tips, ideas and support for how to use the CSL, please check the CATC's Bluetooth Community at:

**<http://bluenews.catc.com/>**

## Contents of this manual

**Chapters 2 to 13** describe the portion of the language that is general to most CATC protocol analyzers.

**Appendix A** lists the specific Bluetooth Input Contexts available for the CATC Bluetooth analyzers.

**Appendix B** contains a sample script file written for a CATC Bluetooth protocol analyzer.



# CHAPTER 2: VALUES

There are five value types that may be manipulated by a script: **integers**, **strings**, **lists**, **raw bytes**, and **null**. CSL is not a strongly typed language. Value types need not be pre-declared. Literals, variables and constants can take on any of the five value types, and the types can be reassigned dynamically.

## Literals

Literals are data that remain unchanged when the program is compiled. Literals are a way of expressing hard-coded data in a script.

### Integers

Integer literals represent numeric values with no fractions or decimal points. Hexadecimal, octal, decimal, and binary notation are supported:

Hexadecimal numbers must be preceded by `0x`: `0x2A`, `0x54`, `0xFFFFFFFF01`

Octal numbers must begin with `0`: `0775`, `017`, `0400`

Decimal numbers are written as usual: `24`, `1256`, `2`

Binary numbers are denoted with `0b`: `0b01101100`, `0b01`, `0b100000`

### Strings

String literals are used to represent text. A string consists of zero or more characters and can include numbers, letters, spaces, and punctuation. An *empty string* (`" "`) contains no characters and evaluates to false in an expression, whereas a non-empty string evaluates to true. Double quotes surround a string, and some standard backslash (`\`) escape sequences are supported.

String	Represented text
<code>"Quote: \"This is a string literal.\""</code>	<code>Quote: "This is a string literal."</code>
<code>"256"</code>	<code>256</code> **Note that this does not represent the integer 256, but only the characters that make up the number.
<code>"abcd!\$%&amp;*"</code>	<code>abcd!\$%&amp;*</code>
<code>"June 26, 2001"</code>	<code>June 26, 2001</code>
<code>"[ 1, 2, 3 ]"</code>	<code>[ 1, 2, 3 ]</code>

**Table 2.1:** Examples of String Literals

## Escape Sequences

These are the available escape sequences in CSL:

Character	Escape Sequence	Example	Output
backslash	\\	"This is a backslash: \\"	This is a backslash: \
double quote	\"	"\"Quotes!\\""	"Quotes!"
horizontal tab	\t	"Before tab\tAfter tab"	Before tab    After tab
newline	\n	"This is how\n to get a newline."	This is how to get a newline.
single quote	\'	"\'Single quote\'"	'Single quote'

**Table 2.2:** Escape Sequences

## Lists

A list can hold zero or more pieces of data. A list that contains zero pieces of data is called an *empty list*. An empty list evaluates to false when used in an expression, whereas a non-empty list evaluates to true. List literals are expressed using the square bracket ([ ]) delimiters. List elements can be of any type, including lists.

```
[1, 2, 3, 4]
[]
["one", 2, "three", [4, [5, [6]]]]
```

## Raw Bytes

Raw binary values are used primarily for efficient access to packet payloads. A literal notation is supported using single quotes:

```
'00112233445566778899AABBCCDDEEFF'
```

This represents an array of 16 bytes with values starting at 00 and ranging up to 0xFF. The values can only be hexadecimal digits. Each digit represents a nybble (four bits), and if there are not an even number of nybbles specified, an implicit zero is added to the first byte. For example:

```
'FFF'
```

is interpreted as

```
'0FFF'
```

## Null

Null indicates an absence of valid data. The keyword `null` represents a literal null value and evaluates to false when used in expressions.

```
result = null;
```

## Variables

Variables are used to store information, or data, that can be modified. A variable can be thought of as a container that holds a value.

All variables have names. Variable names must contain only alphanumeric characters and the underscore ( `_` ) character, and they cannot begin with a number. Some possible variable names are

```
x
_NewValue
name_2
```

A variable is created when it is assigned a value. Variables can be of any value type, and can change type with re-assignment. Values are assigned using the assignment operator ( `=` ). The name of the variable goes on the left side of the operator, and the value goes on the right:

```
x = [ 1, 2, 3 ]
New_value = x
name2 = "Smith"
```

If a variable is referenced before it is assigned a value, it evaluates to null.

There are two types of variables: *global* and *local*.

### Global Variables

Global variables are defined outside of the scope of functions. Defining global variables requires the use of the keyword `set`. Global variables are visible throughout a file (and all files that it includes).

```
set Global = 10;
```

If an assignment in a function has a global as a left-hand value, a variable will not be created, but the global variable will be changed. For example

```
set Global = 10;
Function()
{
    Global = "cat";
    Local = 20;
}
```

will create a local variable called `Local`, which will only be visible within the function `Function`. Additionally, it will change the value of `Global` to `"cat"`, which will be visible to all functions. This will also change its value type from an integer to a string.

## Local Variables

Local variables are not declared. Instead, they are created as needed. Local variables are created either by being in a function's parameter list, or simply by being assigned a value in a function body.

```
Function(Parameter)
{
    Local = 20;
}
```

This function will create a local variable `Parameter` and a local variable `Local`, which has an assigned value of 20.

## Constants

A constant is similar to a variable, except that its value cannot be changed. Like variables, constant names must contain only alphanumeric characters and the underscore ( `_` ) character, and they cannot begin with a number.

Constants are declared similarly to global variables using the keyword `const`:

```
const CONSTANT = 20;
```

They can be assigned to any value type, but will generate an error if used in the left-hand side of an assignment statement later on. For instance,

```
const constant_2 = 3;
Function()
{
    constant_2 = 5;
}
```

will generate an error.

Declaring a constant with the same name as a global, or a global with the same name as a constant, will also generate an error. Like globals, constants can only be declared in the file scope.

# CHAPTER 3: EXPRESSIONS

An expression is a statement that calculates a value. The simplest type of expression is assignment:

```
x = 2
```

The expression `x = 2` calculates 2 as the value of `x`.

All expressions contain operators, which are described in Chapter 4, *Operators*, on page 9. The operators indicate how an expression should be evaluated in order to arrive at its value. For example

```
x + 2
```

says to add 2 to `x` to find the value of the expression. Another example is

```
x > 2
```

which indicates that `x` is greater than 2. This is a Boolean expression, so it will evaluate to either true or false. Therefore, if `x = 3`, then `x > 2` will evaluate to true; if `x = 1`, it will return false.

True is denoted by a non-zero integer (any integer except 0), and false is a zero integer (0). True and false are also supported for lists (an empty list is false, while all others are true), and strings (an empty string is false, while all others are true), and `null` is considered false. However, all Boolean operators will result in integer values.

## select expression

The `select` expression selects the value to which it evaluates based on Boolean expressions. This is the format for a `select` expression:

```
select {
  <expression1> : <statement1>
  <expression2> : <statement2>
  ...
};
```

The expressions are evaluated in order, and the statement that is associated with the first true expression is executed. That value is what the entire expression evaluates to.

```
x = 10
Value_of_x = select {
  x < 5 : "Less than 5";
  x >= 5 : "Greater than or equal to 5";
};
```

The above expression will evaluate to “Greater than or equal to 5” because the first true expression is `x >= 5`. Note that a semicolon is required at the end of a `select` expression because it is not a compound statement and can be used in an expression context.

There is also a keyword `default`, which in effect always evaluates to true. An example of its use is

```
Astring = select {
  A == 1 : "one";
  A == 2 : "two";
  A == 3 : "three";
  A > 3 : "overflow";
  default : null;
};
```

If none of the first four expressions evaluates to true, then `default` will be evaluated, returning a value of `null` for the entire expression.

`select` expressions can also be used to conditionally execute statements, similar to C `switch` statements:

```
select {
  A == 1 : DoSomething();
  A == 2 : DoSomethingElse();
  default: DoNothing();
};
```

In this case the appropriate function is called depending on the value of `A`, but the evaluated result of the `select` expression is ignored.

# CHAPTER 4: OPERATORS

An operator is a symbol that represents an action, such as addition or subtraction, that can be performed on data. Operators are used to manipulate data. The data being manipulated are called *operands*. Literals, function calls, constants, and variables can all serve as operands. For example, in the operation

```
x + 2
```

the variable `x` and the integer `2` are both operands, and `+` is the operator.

## Operations

Operations can be performed on any combination of value types, but will result in a null value if the operation is not defined. Defined operations are listed in the Operand Types column of Table 4.2 on page 11. Any binary operation on a null and a non-null value will result in the non-null value. For example, if

```
x = null;
```

then

```
3 * x
```

will return a value of 3.

A binary operation is an operation that contains an operand on each side of the operator, as in the preceding examples. An operation with only one operand is called a unary operation, and requires the use of a unary operator. An example of a unary operation is

```
!1
```

which uses the logical negation operator. It returns a value of 0.

The unary operators are `sizeof()`, `head()`, `tail()`, `~` and `!`.

## Operator Precedence and Associativity

Operator rules of precedence and associativity determine in what order operands are evaluated in expressions. Expressions with operators of higher precedence are evaluated first. In the expression

```
4 + 9 * 5
```

the `*` operator has the highest precedence, so the multiplication is performed before the addition. Therefore, the expression evaluates to 49.

The associative operator ( ) is used to group parts of the expression, forcing those parts to be evaluated first. In this way, the rules of precedence can be overridden. For example,

$$( 4 + 9 ) * 5$$

causes the addition to be performed before the multiplication, resulting in a value of 65.

When operators of equal precedence occur in an expression, the operands are evaluated according to the associativity of the operators. This means that if an operator's associativity is left to right, then the operations will be done starting from the left side of the expression. So, the expression

$$4 + 9 - 6 + 5$$

would evaluate to 12. However, if the associative operator is used to group a part or parts of the expression, those parts are evaluated first. Therefore,

$$( 4 + 9 ) - ( 6 + 5 )$$

has a value of 2.

In the following table, the operators are listed in order of precedence, from highest to lowest. Operators on the same line have equal precedence, and their associativity is shown in the second column.

Operator Symbol	Associativity
[ ] ( )	Left to right
~ ! sizeof head tail	Right to left
* / %	Left to right
+ -	Left to right
<< >>	Left to right
< > <= >=	Left to right
== !=	Left to right
&	Left to right
^	Left to right
	Left to right
&&	Left to right
	Left to right
=	Right to left

**Table 4.1:** Operator Precedence and Associativity



Operator Symbol	Description	Operand Types	Result Types	Examples
<b>Index Operator</b>				
[ ]	Index or subscript	Raw Bytes	Integer	Raw = '001122' Raw[1] = 0x11
		List	Any	List = [0, 1, 2, 3, [4, 5]] List[2] = 2 List[4] = [4, 5] List[4][1] = 5 *Note: if an indexed Raw value is assigned to any value that is not a byte (> 255 or not an integer), the variable will be promoted to a list before the assignment is performed.
<b>Associative Operator</b>				
( )	Associative	Any	Any	( 2 + 4 ) * 3 = 18 2 + ( 4 * 3 ) = 14
<b>Arithmetic Operators</b>				
*	Multiplication	Integer-integer	Integer	3 * 1 = 3
/	Division	Integer-integer	Integer	3 / 1 = 3
%	Modulus	Integer-integer	Integer	3 % 1 = 0
+	Addition	Integer-integer	Integer	2 + 2 = 4
		String-string	String	"one " + "two" = "one two"
		Raw byte-raw byte	Raw	'001122' + '334455' = '001122334455'
		List-list	List	[1, 2] + [3, 4] = [1, 2, 3, 4]
		Integer-list	List	1 + [2, 3] = [1, 2, 3]
		Integer-string	String	"number = " + 2 = "number = 2" *Note: integer-string concatenation uses decimal conversion.
		String-list	List	"one" + ["two"] = ["one", "two"]
-	Subtraction	Integer-integer	Integer	3 - 1 = 2

Table 4.2: Operators

Operator Symbol	Description	Operand Types	Result Types	Examples
<b>Equality Operators</b>				
<code>==</code>	Equal	Integer-integer	Integer	<code>2 == 2</code>
		String-string	Integer	<code>"three" == "three"</code>
		Raw byte-raw byte	Integer	<code>'001122' == '001122'</code>
		List-list	Integer	<code>[1, [2, 3]] == [1, [2, 3]]</code> *Note: equality operations on values of different types will evaluate to false.
<code>!=</code>	Not equal	Integer-integer	Integer	<code>2 != 3</code>
		String-string	Integer	<code>"three" != "four"</code>
		Raw byte-raw byte	Integer	<code>'001122' != '334455'</code>
		List-list	Integer	<code>[1, [2, 3]] != [1, [2, 4]]</code> *Note: equality operations on values of different types will evaluate to false.
<b>Relational Operators</b>				
<code>&lt;</code>	Less than	Integer-integer	Integer	<code>1 &lt; 2</code>
		String-string	Integer	<code>"abc" &lt; "def"</code>
<code>&gt;</code>	Greater than	Integer-integer	Integer	<code>2 &gt; 1</code>
		String-string	Integer	<code>"xyz" &gt; "abc"</code>
<code>&lt;=</code>	Less than or equal	Integer-integer	Integer	<code>23 &lt;= 27</code>
		String-string	Integer	<code>"cat" &lt;= "dog"</code>
<code>&gt;=</code>	Greater than or equal	Integer-integer	Integer	<code>2 &gt;= 1</code>
		String-string	Integer	<code>"sun" &gt;= "moon"</code> *Note: relational operations on string values are evaluated according to character order in the ASCII table.
<b>Logical Operators</b>				
<code>!</code>	Negation	All combinations of types	Integer	<code>!0 = 1</code> <code>!"cat" = 0</code> <code>!9 = 0</code> <code>!"" = 1</code>
<code>&amp;&amp;</code>	Logical AND	All combinations of types	Integer	<code>1 &amp;&amp; 1 = 1</code> <code>1 &amp;&amp; !"" = 1</code> <code>1 &amp;&amp; 0 = 0</code> <code>1 &amp;&amp; "cat" = 1</code>
<code>  </code>	Logical OR	All combinations of types	Integer	<code>1    1 = 1</code> <code>0    0 = 0</code> <code>1    0 = 1</code> <code>""    !"cat" = 0</code>

Table 4.2: Operators (Continued)

Operator Symbol	Description	Operand Types	Result Types	Examples
<b>Bitwise Logical Operators</b>				
<code>~</code>	Bitwise complement	Integer-integer	Integer	<code>~0b11111110 = 0b00000001</code>
<code>&amp;</code>	Bitwise AND	Integer-integer	Integer	<code>0b11111110 &amp; 0b01010101 = 0b01010100</code>
<code>^</code>	Bitwise exclusive OR	Integer-integer	Integer	<code>0b11111110 ^ 0b01010101 = 0b10101011</code>
<code> </code>	Bitwise inclusive OR	Integer-integer	Integer	<code>0b11111110   0b01010101 = 0b11111111</code>
<b>Shift Operators</b>				
<code>&lt;&lt;</code>	Left shift	Integer-integer	Integer	<code>0b11111110 &lt;&lt; 3 = 0b11110000</code>
<code>&gt;&gt;</code>	Right shift	Integer-integer	Integer	<code>0b11111110 &gt;&gt; 1 = 0b01111111</code>
<b>Assignment Operator</b>				
<code>=</code>	Assignment	Any	Any	<code>A = 1</code> <code>B = C = A</code>
<b>List Operators</b>				
<code>sizeof()</code>	Number of elements	Any	Integer	<code>sizeof([1, 2, 3]) = 3</code> <code>sizeof('0011223344') = 5</code> <code>sizeof("string") = 6</code> <code>sizeof(12) = 1</code> <code>sizeof([1, [2, 3]]) = 2</code> *Note: the last example demonstrates that the <code>sizeof()</code> operator returns the shallow count of a complex list.
<code>head()</code>	Head	List	Any	<code>head([1, 2, 3]) = 1</code> *Note: the Head of a list is the first item in the list.
<code>tail()</code>	Tail	List	List	<code>tail([1, 2, 3]) = [2, 3]</code> *Note: the Tail of a list includes everything except the Head.

**Table 4.2: Operators (Continued)**



# CHAPTER 5: COMMENTS

Comments may be inserted into scripts as a way of documenting what the script does and how it does it. Comments are useful as a way to help others understand how a particular script works. Additionally, comments can be used as an aid in structuring the program.

Comments in CSL begin with a hash mark (#) and finish at the end of the line. The end of the line is indicated by pressing the Return or Enter key. Anything contained inside the comment delimiters is ignored by the compiler. Thus,

```
# x = 2;
```

is not considered part of the program. CSL supports only end-of-line comments, which means that comments can be used only at the end of a line or on their own line. It's not possible to place a comment in the middle of a line.

Writing a multi-line comment requires surrounding each line with the comment delimiters

```
# otherwise the compiler would try to interpret  
# anything outside of the delimiters  
# as part of the code.
```

The most common use of comments is to explain the purpose of the code immediately following the comment. For example:

```
# Add a profile if we got a server channel  
if(rfChannel != "Failure")  
{  
    result = SDPAddProfileServiceRecord(rfChannel, "ObjectPush");  
    Trace("SDPAddProfileServiceRecord returned ", result, "\n");  
}
```



# CHAPTER 6: KEYWORDS

Keywords are reserved words that have special meanings within the language. They cannot be used as names for variables, constants or functions.

In addition to the operators, the following are keywords in CSL:

Keyword	Usage
select	select expression
set	define a global variable
const	define a constant
return	return statement
while	while statement
for	for statement
if	if statement
else	if-else statement
default	select expression
null	null value
in	input context
out	output context

**Table 6.1:** Keywords





# CHAPTER 7: STATEMENTS

Statements are the building blocks of a program. A program is made up of list of statements.

Seven kinds of statements are used in CSL: expression statements, if statements, if-else statements, while statements, for statements, return statements, and compound statements.

## Expression Statements

An expression statement describes a value, variable, or function.

```
<expression>;
```

Here are some examples of the different kinds of expression statements:

```
Value: x + 3;
Variable: x = 3;
Function: Trace ( x + 3 );
```

The variable expression statement is also called an *assignment statement*, because it assigns a value to a variable.

## if Statements

An if statement follows the form

```
if <expression> <statement>
```

For example,

```
if ( 3 && 3 ) Trace("True!");
```

will cause the program to evaluate whether the expression `3 && 3` is nonzero, or True. It is, so the expression evaluates to True and the `Trace` statement will be executed. On the other hand, the expression `3 && 0` is not nonzero, so it would evaluate to False, and the statement wouldn't be executed.

## if-else Statements

The form for an if-else statement is

```
if <expression> <statement1>
else <statement2>
```

The following code

```
if ( 3 - 3 || 2 - 2 ) Trace ( "Yes" );
else Trace ( "No" );
```

will cause “No” to be printed, because  $3 - 3 \ || \ 2 - 2$  will evaluate to False (neither  $3 - 3$  nor  $2 - 2$  is nonzero).

## while Statements

A while statement is written as

```
while <expression> <statement>
```

An example of this is

```
x = 2;
while ( x < 5 )
{
    Trace ( x, " , " );
    x = x + 1;
}
```

The result of this would be


```
2, 3, 4,
```

## for Statements

A for statement takes the form

```
for ( <expression1>; <expression2>; <expression3> ) <statement>
```

The first expression initializes, or sets, the starting value for  $x$ . It is executed one time, before the loop begins. The second expression is a conditional expression. It determines whether the loop will continue -- if it evaluates true, the function keeps executing and proceeds to the statement; if it evaluates false, the loop ends. The third expression is executed after every iteration of the statement.



**Figure 7-1:** Execution of a for statement

The example

```
for ( x = 2; x < 5; x = x + 1 ) Trace ( x, "\n" );
```

would output

```

2
3
4

```

The example above works out like this: the expression `x = 2` is executed. The value of `x` is passed to `x < 5`, resulting in `2 < 5`. This evaluates to true, so the statement `Trace ( x, "\n" )` is performed, causing 2 and a new line to print. Next, the third expression is executed, and the value of `x` is increased to 3. Now, `x < 5` is executed again, and is again true, so the `Trace` statement is executed, causing 3 and a new line to print. The third expression increases the value of `x` to 4; `4 < 5` is true, so 4 and a new line are printed by the `Trace` statement. Next, the value of `x` increases to 5. `5 < 5` is *not* true, so the loop ends.

## return Statements

Every function returns a value, which is usually designated in a `return` statement. A `return` statement returns the value of an expression to the calling environment. It uses the following form:

```
return <expression>;
```

An example of a `return` statement and its calling environment is

```
Trace ( HiThere() );
...
HiThere()
{
    return "Hi there";
}
```

The call to the primitive function `Trace` causes the function `HiThere()` to be executed. `HiThere()` returns the string “Hi there” as its value. This value is passed to the calling environment (`Trace`), resulting in this output:

```
Hi there
```

A `return` statement also causes a function to stop executing. Any statements that come after the `return` statement are ignored, because `return` transfers control of the program back to the calling environment. As a result,

```
Trace ( HiThere() );
...
HiThere()
{
    a = "Hi there";
    return a;
    b = "Goodbye";
    return b;
}
```

will output only

```
Hi there
```

because when `return a;` is encountered, execution of the function terminates, and the second `return` statement (`return b;`) is never processed. However,

```
Trace ( HiThere() );  
...  
HiThere()  
{  
    a = "Hi there";  
    b = "Goodbye";  
    if ( 3 != 3 ) return a;  
    else return b;  
}
```

will output

```
Goodbye
```

because the `if` statement evaluates to false. This causes the first `return` statement to be skipped. The function continues executing with the `else` statement, thereby returning the value of `b` to be used as an argument to `Trace`.

## Compound Statements

A compound statement, or *statement block*, is a group of one or more statements that is treated as a single statement. A compound statement is always enclosed in curly braces ( `{ }` ). Each statement within the curly braces is followed by a semicolon; however, a semicolon is not used following the closing curly brace.

The syntax for a compound statement is

```
{  
    <first_statement>;  
    <second_statement>;  
    ...  
    <last_statement>;  
}
```

An example of a compound statement is

```
{  
    x = 2;  
    x + 3;  
}
```

It's also possible to nest compound statements, like so:

```
{  
    x = 2;  
    {  
        y = 3;  
    }  
    x + 3;  
}
```

Compound statements can be used anywhere that any other kind of statement can be used.

```
if (3 && 3)
{
    result = "True!";
    Trace(result);
}
```

Compound statements are required for function declarations and are commonly used in `if`, `if-else`, `while`, and `for` statements.



## CHAPTER 8: PREPROCESSING

The preprocessing command `%include` can be used to insert the contents of a file into a script. It has the effect of copying and pasting the file into the code. Using `%include` allows the user to create modular script files that can then be incorporated into a script. This way, commands can easily be located and reused.

The syntax for `%include` is this:

```
%include "includefile.inc"
```

The quotation marks around the filename are required, and by convention, the included file has a `.inc` extension.

The filenames given in the include directive are always treated as being relative to the current file being parsed. So, if a file is referenced via the preprocessing command in a `.dec` file, and no path information is provided (`%include "file.inc"`), the application will try to load the file from the current directory. Files that are in a directory one level up from the current file can be referenced using `..\file.inc`, and likewise, files one level down can be referenced using the relative pathname (`directory\file.inc`). Last but not least, files can also be referred to using a full pathname, such as `"C:\global_scripts\include\file.inc"`.

If a file is not found in current directory, application will try to find it in shared directory `Scripts\Shared\`.





## CHAPTER 9: CONTEXT

The context is the mechanism by which transaction data is passed in and out of the scripts. There is an output context that is modified by the script, and there are possibly multiple input contexts that the script will be invoked on separately.

A context serves two roles: firstly, it functions as a symbol table whose values are local to a particular transaction; secondly, it functions as an interface to the application. Two keywords are used to reference symbols in the context: `in` and `out`. Dot notation is used to specify a symbol within a context:

```
out.symbol = "abcd";  
out.type = in.type;
```

The output context can be read and written to, but the input context can only be read. Context symbols follow the same rules as local variables: they are created on demand, and uninitialized symbols always evaluate to null.

When a script is first invoked, it is given an input context that corresponds to a packet or transaction that is a candidate for being a part of a larger transaction. The output context is initially empty. It is the script's job to examine the input context and decide if it qualifies for membership in the type of transaction that the script was designed to decode. If it qualifies, the appropriate values will be decoded and put in the output context symbol table, and if the transaction is complete, it will be done. If the transaction is not complete, the script will indicate this to the application based on its return value, and will be invoked again with the same output context, but a new input context. The script then must decide if this new input context is a member of the transaction, and keep doing this until the transaction is complete.

In order to accomplish all this, state information should be placed in the output context. It should be possible to use the output context of one transaction as an input context to another transaction.



# CHAPTER 10: FUNCTIONS

A function is a named statement or a group of statements that are executed as one unit. All functions have names. Function names must contain only alphanumeric characters and the underscore ( `_` ) character, and they cannot begin with a number.

A function can have zero or more *parameters*, which are values that are passed to the function statement(s). Parameters are also known as *arguments*. Value types are not specified for the arguments or return values. Named arguments are local to the function body, and functions can be called recursively.

The syntax for a function declaration is

```
name(<parameter1>, <parameter2>, ...)  
{  
    <statements>  
}
```

The syntax to call a function is

```
name(<parameter1>, <parameter2>, ...)
```

So, for example, a function named `add` can be declared like this:

```
add(x, y)  
{  
    return x + y;  
}
```

and called this way:

```
add(5, 6);
```

This would result in a return value of 11.

Every function returns a value. The return value is usually specified using a `return` statement, but if no `return` statement is specified, the return value will be the value of the last statement executed.

Arguments are not checked for appropriate value types or number of arguments when a function is called. If a function is called with fewer arguments than were defined, the specified arguments are assigned, and the remaining arguments are assigned to null. If a function is called with more arguments than were defined, the extra arguments are ignored. For example, if the function `add` is called with just one argument

```
add(1);
```

the parameter `x` will be assigned to 1, and the parameter `y` will be assigned to null, resulting in a return value of 1. But if `add` is called with more than two arguments

```
add(1, 2, 3);
```

`x` will be assigned to 1, `y` to 2, and 3 will be ignored, resulting in a return value of 3.

All parameters are passed by value, not by reference, and can be changed in the function body without affecting the values that were passed in. For instance, the function

```
add_1(x, y)
{
    x = 2;
    y = 3;
    return x + y;
}
```

reassigns parameter values within the statements. So,

```
a = 10;
b = 20;
add_1(a, b);
```

will have a return value of 5, but the values of a and b won't be changed.

The scope of a function is the file in which it is defined (as well as included files), with the exception of primitive functions, whose scopes are global.

Calls to undefined functions are legal, but will always evaluate to null and result in a compiler warning.

# CHAPTER 11: PRIMITIVES

Primitive functions are called similarly to regular functions, but they are implemented outside of the language. Some primitives support multiple types for certain arguments, but in general, if an argument of the wrong type is supplied, the function will return null.

## Call( )

```
Call( <function_name string>, <arg_list list> )
```

Parameter	Meaning	Default Value	Comments
function_name <i>string</i>			
arg_list <i>list</i>			Used as the list of parameters in the function call.

### *Return value*

Same as that of the function that is called.

### *Comments*

Calls a function whose name matches the `function_name` parameter. All scope rules apply normally. Spaces in the `function_name` parameter are interpreted as the ‘\_’ (underscore) character since function names cannot contain spaces.

### *Example*

```
Call("Format", ["the number is %d", 10])
```

is equivalent to:

```
Format("the number is %d", 10)
```

## Format( )

```
Format (<format string>, <value string or integer>)
```

Parameter	Meaning	Default Value	Comments
format <i>string</i>			
value <i>string</i> or <i>integer</i>			

### *Return value*

None.

### Comments

`Format` is used to control the way that arguments will print out. The format string may contain conversion specifications that affect the way in which the arguments in the value string are returned. Format conversion characters, flag characters, and field width modifiers are used to define the conversion specifications.

### Example

```
Format("0x%02X", 20);
```

would yield the string `0x14`.

`Format` can only handle one value at a time, so

```
Format("%d %d", 20, 30);
```

would not work properly. Furthermore, types that do not match what is specified in the format string will yield unpredictable results.

## Format Conversion Characters

These are the format conversion characters used in CSL:

Code	Type	Output
<code>c</code>	Integer	Character
<code>d</code>	Integer	Signed decimal integer.
<code>i</code>	Integer	Signed decimal integer
<code>o</code>	Integer	Unsigned octal integer
<code>u</code>	Integer	Unsigned decimal integer
<code>x</code>	Integer	Unsigned hexadecimal integer, using "abcdef."
<code>X</code>	Integer	Unsigned hexadecimal integer, using "ABCDEF."
<code>s</code>	String	String

**Table 11.1:** Format Conversion Characters

A conversion specification begins with a percent sign (%) and ends with a conversion character. The following optional items can be included, in order, between the % and the conversion character to further control argument formatting:

- Flag characters are used to further specify the formatting. There are five flag characters:
  - A minus sign (-) will cause an argument to be left-aligned in its field. Without the minus sign, the default position of the argument is right-aligned.
  - A plus sign will insert a plus sign (+) before a positive signed integer. This only works with the conversion characters `d` and `i`.
  - A space will insert a space before a positive signed integer. This only works with the conversion characters `d` and `i`. If both a space and a plus sign are used, the space flag will be ignored.

- A hash mark (#) will prepend a 0 to an octal number when used with the conversion character o. If # is used with x or X, it will prepend 0x or 0X to a hexadecimal number.
- A zero (0) will pad the field with zeros instead of with spaces.
- Field width specification is a positive integer that defines the field width, in spaces, of the converted argument. If the number of characters in the argument is smaller than the field width, then the field is padded with spaces. If the argument has more characters than the field width has spaces, then the field will expand to accommodate the argument.

## GetNBits ( )

GetNBits (<bit\_source list or raw>, <bit\_offset integer>, <bit\_count integer>)

Parameter	Meaning	Default Value	Comments
bit_source list, raw, or integer			Can be an integer value (4 bytes) or a list of integers that are interpreted as bytes.
bit_offset integer	Index of bit to start reading from		
bit_count integer	Number of bits to read		

### Return value

integer, list or raw.

### Comments

Reads bit\_count bits from bit\_source starting at bit\_offset. Will return null if bit\_offset + bit\_count exceeds the number of bits in bit\_source. If bit\_count is 32 or less, the result will be returned as an integer. Otherwise, the result will be returned in a list format that is the same as the input format. GetNBits also sets up the bit data source and global bit offset used by NextNBits and PeekNBits. Note that bits are indexed starting at bit 0.

### Example

```
raw = 'FOFO';           # 1111000011110000 binary
result = GetNBits ( raw, 2, 4 );
Trace ( "result = ", result );
```

The output would be

```
result = C             # The result is given in hexadecimal. The result
in binary is 1100.
```

In the call to GetNBits: starting at bit 2, reads 4 bits (1100), and returns the value 0xC.

## GetBitOffset()

GetBitOffset()

Parameter	Meaning	Default Value	Comments
N/A			

### *Return value*

integer

### *Comments*

Returns the current bit offset that is used in `NextNBits` or `PeekNBits`.

### *Example*

```
raw = 'FOFO';# 1111000011110000 binary
result1 = GetNBits ( raw, 2, 4 );
result2 = PeekNBits(5);
result3 = NextNBits(2);
Trace ( "Offset = ", GetBitOffset() );
```

The example generates this Trace output:

```
Offset = D
```

## PeekNBits()

PeekNBits(<bit\_count integer>)

Parameter	Meaning	Default Value	Comments
bit_count integer			

### *Return value*

integer, list or raw.

### *Comments*

Reads `bit_count` bits from the data source. The difference between `PeekNBits` and `NextNBits` is that `PeekNBits` does not advance the global bit offset. `PeekNBits` can be used to make decisions about how to parse the next fields without affecting subsequent calls to `NextNBits`. If `PeekNBits` is called without a prior call to `GetNBits`, the result is undefined. Note that bits are indexed starting at bit 0.

### *Example*

```
raw = 'FOFO';# 1111000011110000 binary
result1 = GetNBits ( raw, 2, 4 );
result2 = PeekNBits(5);
```



```

result3 = NextNBits(2);
Trace ( "result1 = ", result1, "result2 = ", result2, "result3 = ",
result3 );

```

This will generate this Trace output:

```
result1 = C result2 = 7 result3 = 0
```

In the call to `GetNBits`: starting at bit 2, reads 4 bits (1100), and returns the value 0xC.

In the call to `PeekNBits`: starting at bit 6, reads 5 bits (00111), and returns the value 0x7.

In the call to `NextNBits`: starting at bit 6, reads 2 bits (00), and returns the value 0x0.

## NextNBits ( )

```
NextNBits (<bit_count integer>)
```

Parameter	Meaning	Default Value	Comments
<code>bit_count integer</code>			

### *Return value*

integer, list or raw.

### *Comments*

Reads `bit_count` bits from the data source specified in the last call to `GetNBits`, starting after the last bit that the previous call to `GetNBits` or `NextNBits` returned. If called without a previous call to `GetNBits`, the result is undefined. Note that bits are indexed starting at bit 0.

### *Example*

```

raw = 'FOFO';# 1111000011110000 binary
result1 = GetNBits ( raw, 2, 4 );
result2 = NextNBits(5);
result3 = NextNBits(2);
Trace ( "result1 = ", result1, "result2 = ", result2, "result3 = ",
result3 );

```

This will generate this trace output:

```
result1 = C result2 = 7 result3 = 2
```

In the call to `GetNBits`: starting at bit 2, reads 4 bits (1100), and returns the value 0xC.

In the first call to `NextNBits`: starting at bit 6, reads 5 bits (00111), and returns the value 0x7.

In the second call to `NextNBits`: starting at bit 11 (= 6 + 5), reads 2 bits (10), and returns the value 0x2.

## Resolve( )

`Resolve( <symbol_name string> )`

Parameter	Meaning	Default Value	Comments
<code>symbol_name string</code>			

### *Return value*

The value of the symbol. Returns null if the symbol is not found.

### *Comments*

Attempts to resolve the value of a symbol. Can resolve global, constant and local symbols. Spaces in the `symbol_name` parameter are interpreted as the ‘\_’ (underscore) character since symbol names cannot contain spaces.

### *Example*

```
a = Resolve( "symbol" );
```

is equivalent to:

```
a = symbol;
```

## Trace( )

`Trace( <arg1 any>, <arg2 any>, ... )`

Parameter	Meaning	Default Value	Comments
<code>arg any</code>			The number of arguments is variable.

### *Return value*

None.

### *Comments*

The values given to this function are given to the debug console.

### *Example*

```
list = ["cat", "dog", "cow"]
Trace("List = ", list, "\n");
```

would result in the output

```
List = [cat, dog, cow]
```





# CHAPTER 12: DECODER PRIMITIVES

## Abort ( )

Abort ( )

Parameter	Meaning	Default Value	Comments
N/A			

### *Return value*

An integer that should be passed back to the application unchanged.

### *Comments*

Called when an input context renders the currently pending transaction done, but is not itself a member of that transaction. An example would be an input transaction that represents some sort of reset condition that renders all pending transactions invalid. The input transaction is not consumed by this action and will go on to be considered for other pending transactions.

### *Example*

```
if ( IsReset )
return Abort();
```

## AddCell ( )

AddCell(<name *string*>, <value *string*>, <description *string* or *null*>, <color *integer* or *list*>, <additional\_info *any*>)

Parameter	Meaning	Default Value	Comments
name <i>string</i>			Displays in the name field of the cell.
value <i>string</i>			Displays in the value field of the cell.
description <i>string</i> or <i>null</i>			Displays in tool tip.
color <i>integer</i> or <i>list</i>		If not specified, a default color is used	Color can be specified as either a packed color value in an integer, or as an array of RGB values ranging from 0-255. Displays in the name field of the cell.

Parameter	Meaning	Default Value	Comments
<code>additional_info any</code>			Used to create special cells or to modify cell attributes. The values are predefined constants, and zero or more of them may be used at one time. Possible values are: _COLLAPSED _ERROR _EXPANDED [_FIXEDWIDTH, w] _HIDDEN _MONOCOLOR _MONOFIELD _SHOWN (default) _WARNING

### *Return value*

None.

### *Comments*

Adds a display cell to the current output context. Cells are displayed in the order that they are added. The name and value strings are displayed directly in the cell.

### *Example*

```
# Create a regular cell named Normal with a value "Cell" and tool tip
"Normal cell":
AddCell( "Normal", "Value1", "Normal cell" );

# Use the _MONOCOLOR value in the additional_info parameter to create a
cell with a color value of 0x881122 in both the name and value fields:
AddCell( "MonoColor", "Value2", "MonoColor cell", 0x881122, _MONOCOLOR
);

# Use the _MONOFIELD value to create a cell with only a name field:
AddCell( "MonoField", "Value3", "MonoField cell", [255, 200, 200],
_MONOFIELD );

# Use the _ERROR value to create a cell with a red value field:
AddCell( "Error", "Value4", "Error cell", 0xcc1155, _ERROR );

# Use the _WARNING value to create a cell with a yellow value field:
AddCell( "Warning", "Value5", "Warning cell", 0x00BB22, _WARNING );

# Use the [_FIXEDWIDTH, w] value to create a cell with a fixed width of
20 in conjunction with the error value to create a fixed width cell with
a red value field:
AddCell( "Fixed Width 20", "Value6", "Fixed Width and Error cell",
0x001122, [_FIXEDWIDTH, 20], _ERROR );
```

The output of the example is:

Normal	MonoColor	MonoField	Error	Warning	Fixed Width 20
Value1	Value2		Value4	Value5	Value6

Figure 12-1: Example output for AddCell

## AddDataCell( )

AddDataCell(<data\_value raw, list or integer>, <additional\_info any>, ...)

Parameter	Meaning	Default Value	Comments
data_value raw, list, or integer			Interpreted the same way as GetNBits interprets data_source
additional_info any			Used to create special cells or to modify cell attributes. Possible values are: _BYTES _COLLAPSED _DWORDS _EXPANDED _HIDDEN _SHOWN (default)

### Return value

None.

### Comments

Creates an expandable/collapsible cell for viewing raw data such as data payloads. Data can be raw bytes, an integer, or a list. If an integer is used, it will be interpreted as 4 bytes of data. Specifying `_BYTES` or `_DWORDS` in an `additional_info` field will force data to be interpreted as bytes or quadlets. `_COLLAPSED`, `_EXPANDED`, `_HIDDEN` and `_SHOWN` are all interpreted the same as in a regular `AddCell` call.

### Example

```
# Creates a data cell with 2 dwords (32-bit integers) of data.
AddDataCell( '0123456789ABCDEF', _DWORDS );

# Creates a data cell with 4 bytes. Integer data values are always
interpreted as 32 bits of data.
AddDataCell( 0x11223344, _BYTES );
```

The output of the example is:

Test Cells	Data	Data
0	01234567 89ABCDEF	11 22 33 44
Test Cells	Data	Data
1	2 quadlets	4 bytes

Figure 12-2: Example output for AddDataCell

## AddEvent ( )

AddEvent(<Group *string*>, <Value *string*> )

Parameter	Meaning	Default Value	Comments
Group <i>string</i>	The name of the group		Corresponds to the name of a field that might be encountered while decoding.
Value <i>string</i>	A value that will be associated with the group		Corresponds to a field value that might be encountered while parsing.

### Return value

None.

### Comments

Events are used for transaction searching and for transaction summary. This function is only effective when called during the `ProcessData()` phase of decoding. Event groups and values are stored globally for transaction levels and new ones are created as they are encountered. Each transaction contains information as to which events were associated with it.

### Example

```
AddEvent( "DataLength", Format( "%d",
out.DataLength ));
```



## AddSeparator( )

AddSeparator(<additional\_info any>, ...)

Parameter	Meaning	Default Value	Comments
additional_info any			Used to create special cells or to modify cell attributes. The values are predefined constants. Possible values are: _COLLAPSED _EXPANDED _HIDDEN _SHOWN (default)

### Return value

None.


### Comments

Creates a separator cell. \_COLLAPSED, \_EXPANDED, \_HIDDEN, and \_SHOWN are all interpreted the same as in a regular AddCell call.

### Example

```
AddCell( "Stuff", "Things" );
# AddSeparator adds a space between the previous and subsequent cells.
AddSeparator();
AddCell( "More stuff", "More things" );
```

The output of the example is:



**Figure 12-3:** Separator cell example

## BeginCellBlock( )

BeginCellBlock(<name string>, <value string>, <description string or null>, <color integer or list>, <additional\_info any>)

Parameter	Meaning	Default Value	Comments
name string			Displays in the name field of the cell.
value string			Displays in the value field of the cell.
description string or null			Displays in tool tip.

Parameter	Meaning	Default Value	Comments
color <i>integer</i> or <i>list</i>		If not specified, a default color is used	Color can be specified as either a packed color value in an integer, or as an array of RGB values ranging from 0-255. Displays in the name field of the cell.
additional_info <i>any</i>			Used to create special cells or to modify cell attributes. The values are predefined constants, and zero or more of them may be used at one time. Possible values are: [_BLOCKNAME, x] _COLLAPSED _ERROR _EXPANDED [_FIXEDWIDTH, w] _HIDDEN _MONOCOLOR _MONOFIELD _SHOWN (default) _WARNING

### Return value

None.

### Comments

Begins a cell block and adds a block header cell. This is a special cell that can be collapsed and expanded. The collapsed/expanded state of this cell affects cells in the group according to their `_COLLAPSED`, `_EXPANDED` attributes. All calls to `AddCell` after a call to `BeginCellBlock()` will put the new cells into this group until a call to `EndCellBlock` is made.

Cell blocks can be nested.

### Example

```
# Begin the 'red' group. For clarity these cells will be red:
BeginCellBlock( "Red Group", null, null, 0x0000ff, _MONOFIELD );

# This cell will be displayed when the red group is in the expanded
state:
AddCell( "Red is", "Expanded", null, 0x0000ff, _EXPANDED );

# This cell will be displayed when the red group is collapsed:
AddCell( "Red is", "Collapsed", null, 0x0000ff, _COLLAPSED );

# This begins the nested blue group. Nothing in the blue group will be
displayed unless the red group is expanded:
BeginCellBlock( "Blue Group", null, null, 0xff0000, _MONOFIELD,
_EXPANDED, [_BLOCKNAME, "BlockName"] );

# This cell is only displayed when the blue group is visible and
expanded:
AddCell( "Blue is", "Expanded", null, 0xff0000, _EXPANDED );
```

```

# This cell is also only displayed when the blue group is visible and
expanded:
AddCell( "Blue", "Too", null, 0xff0000, _EXPANDED );
# This cell is only displayed when the blue group is visible and
collapsed:
AddCell( "Blue is", "Collapsed", null, 0xff0000, _COLLAPSED );
# This ends the blue group.
EndCellBlock();
# Cells with the _SHOWN attribute are always displayed. This is the
default:
AddCell( "Always", "Shown", null, 0x0000ff, _SHOWN );
# This cell will never be displayed. In a real script this would be
driven by a variable:
AddCell( "Never", "Shown", null, 0x0000ff, _HIDDEN );
# This ends the red group.
EndCellBlock();

```

The output of the example is:

Red Group	Red is	Always
	Collapsed	Shown

**Figure 12-4:** Example output for BeginCellBlock with red group collapsed

Red Group	Red is	Blue Group	Blue is	Always
	Expanded		Collapsed	Shown

**Figure 12-5:** Example output for BeginCellBlock with red group expanded and blue group collapsed

Red Group	Red is	Blue Group	Blue is	Blue	Always
	Expanded		Expanded	Too	Shown

**Figure 12-6:** Example output for BeginCellBlock with red group expanded and blue group expanded

## Complete()

Complete()

Parameter	Meaning	Default Value	Comments

*Return value*

An integer that should be passed back to the application unchanged.

*Comments*

This should be called when it has been decided that an input context has been accepted into a transaction, and that the transaction is complete. The return value of this function should be passed back to the application from the `ProcessData` function. This function could be used to associate the input context with the output context.

*Example*

```
if ( done )
    return Complete();
```

**EndCellBlock( )**

EndCellBlock( )

Parameter	Meaning	Default Value	Comments
-----------	---------	---------------	----------

*Return value*

None.

*Comments*

Ends a cell block that was started with `BeginCellBlock()`.

*Example*

See `BeginCellBlock()`.

**Pending( )**

Pending( )

Parameter	Meaning	Default Value	Comments
-----------	---------	---------------	----------

*Return value*

An integer that should be passed back to the application unchanged.

*Comments*

This should be called when it has been decided that an input context has been accepted into a transaction, but that the transaction still requires further input to be complete. This function could be used to associate input contexts with the output context. The return value of this function should be returned to the application in the `ProcessData` function.

*Example*

```
if ( done )
  return Complete();
else return Pending();
```

**Reject ( )**

`Reject()`

Parameter	Meaning	Default Value	Comments

*Return value*

An integer that should be passed back to the application unchanged.

*Comments*

Called when it is decided that the input context does not meet the criteria for being a part of the current transaction. The output context should not be modified before this decision is made. The return value of this function should be returned by the `ProcessData` function.

*Example*

```
if ( UnknownValue )
  return Reject();
```



# CHAPTER 13: MODULES

Modules are a collection of functions and global data dedicated to decoding a certain type of transaction. Each module consists of one primary file (.dec), and possibly several included files (.inc).

## Module Functions

Three functions are used as entry-points into a decoding module. They are called by the application and are used both in the initial transaction decoding phase, and each time that a transaction needs to be displayed.

### **ProcessData ( )**

Called repeatedly with input contexts representing transactions of the specified input types. Decides if input transaction is a member of this transaction, or if it begins a new transaction. This function will be called first using incomplete output transactions. If the input transaction is not accepted into any of the pending transactions, it will be called with an empty output transaction to see if it starts a new transaction.

### **CollectData ( )**

Called with each input transaction that was previously accepted by the function `ProcessData`. Generates all output context data that would be required for input into a higher level transaction.

### **BuildCellList ( )**

Called with the output context generated by the call to `CollectData`, and no input context. This function is responsible for adding display cells based on the data collected by `CollectData`.

Note that there is some flexibility in the use of these functions. For example, if it is easier for a particular protocol to build cells in `CollectData`, cells could be generated there, and `BuildCellList` could be left empty. Another approach would be to have `ProcessData` do everything (generate output data, and build cell lists) and then implement `CollectData` as a pass-thru to `ProcessData`. This will be less efficient in the decoding phase but may reduce some repetition of code. These decisions are dependent on the protocol to be decoded.

## Module Data

There are several standard global variables that should be defined in a module which are queried by the application to figure out what the module is supposed to do.

### ModuleType

Required. A string describing the role of the script. Currently, only `Transaction Decoder` and `DataBlock Decoder` are valid.

#### *Example*

```
set ModuleType = "Transaction Decoder";
```

`Transaction Decoder` **USES** `ProcessData()`. `DataBlock Decoder` **does not**.

### OutputType

Required. A string label describing the output of the script. Example : `AVC Transaction`

#### *Example*

```
set OutputType = "AV/C Transaction";
```

### InputType

Required. A string label describing the input to the script. Input and output types should be matched by the application in order to decide which modules to invoke on which contexts.

#### *Example*

```
set InputType = "1394 Transaction";
```

### LevelName

Optional. A string that names this decoder.

#### *Example*

```
set LevelName = "AV/C Test Transactions";
```

### DecoderDesc

Optional. A string that describes this decoder. Displays as a toolbar icon tool tip.

#### *Example*

```
set DecoderDesc = "View test transactions";
```



**Icon**

Optional. File name of an icon to display on the toolbar. Must be a 19x19 pixel bitmap file.

*Example*

```
set Icon = "bitmap.bmp";
```

**LelelColor**

Optional. Specifies color for Transaction Number Cell.

*Example*

```
set LelelColor= 0x105548;
```



# APPENDIX A: BLUETOOTH DATA

The Merlin application decodes several layers of the Bluetooth protocol and provides input context as follows:

## Packet Level

in.Data packet)	- data block (packet payload) (null if no data in packet)
in.DataLength	- length of packet payload (null if no data in packet)
in.ScoData	- SCO data block (voice) (null if no SCO data in packet)
in.ScoDataLength	- length of SCO data (null if no SCO data in packet)
in.Slave	- 1 - Slave/ 0 - Master
in.AmAddr	- Am address
in.Type	- type of packet
in.Flow	- packet flow bit
in.Seqn	- packet seqn bit
in.L_CH	- packet L_CH value

## L2CAP

in.Data	- L2CAP data block
in.DataLength	- length of data block
in.Slave	- 1 - Slave/ 0 - Master
in.AmAddr	- Am address
in.Cid	- L2CAP CID value

## RFCOMM

in.Data	- RFCOMM data block
in.DataLength	- length of data block
in.Slave	- 1 - Slave/ 0 - Master
in.AmAddr	- Am address
in.Dlci	- RFCOMM dlci value

## HDLC and PPP

in.Data	- HDLC data block
in.DataLength	- length of data block
in.Protocol	- PPP protocol value
in.Slave	- 1-Slave/ 0-\Master
in.AmAddr	- Am address



# APPENDIX B: SAMPLE BNEP SCRIPT

## Sample Script

```
#
# Copyright (c) 2001, Computer Access Technology Corp. - All Rights Reserved
#
# BNEP.dec
#
# BNEP Transaction Decoder
#

#include "BTDecoderTools.inc" #include usefull decoder functions
#include "ethernet.inc" #ethernet related tables

const Bnep_color = 0x551234;

set ModuleType = "Transaction Decoder";
set OutputType = "BNEP";
set InputType = "L2CAP";
set LevelName = "BNEP";
set LevelColor = Bnep_color;
set Icon = "bnep.bmp"; #it's possible to specify toolbar icon
set DecoderDesc = "View Bluetooth Network Encapsulation Protocol Layer";

#---- L2CAP Input Data
# in.Data - Data_block to decode
# in.DataLength - Length of Data_block
# in.Slave - Slave or Master
# in.AmAddr - AmAddr
# in.Cid - cid
#
#---- Output
#
```

```
# out.Data      - Data block
# out.DataLength - Length of Data_block
# out.EtherProtocol - General Ethernet protocol

#
# Parsing entry point
#

ProcessData()
{
    type = GetNBits(in.Data, 1, 7);
    select
    {
        type == 0: AddEvent("BNEP Type", "GENERAL_ETHERNET (0x00)");
        type == 1: AddEvent("BNEP Type", "CONTROL (0x01)");
        type == 2: AddEvent("BNEP Type", "COMPR_ETHERNET (0x02)");
        type == 3: AddEvent("BNEP Type", "CMPR_ETH_SRC_ONLY (0x03)");
        type == 4: AddEvent("BNEP Type", "CMPR_ETH_DST_ONLY (0x04)");
        default: AddEvent("BNEP Type", Format("Reserved - 0x%02X", type) );
    };
    return Complete();    # transaction is finished
}

set Type = null;
set DecodedBytes = 0;
set E = 0;           # extension flag
set Bnep_Error = 0;

CollectData()
{
    out.EtherProtocol = null;
    out.Slave = in.Slave;
    out.AmAddr = in.AmAddr;
    ShowAmAddrSlave();
}
```

```
Bnep_Error = 0;

byte = GetNBits(in.Data, 0, 8);
E = (byte >> 7) & 1;
AddCell("E",Format ("%d", E),"Extension flag", Bnep_color);
```

```
DecodedBytes = 1;
Type = byte & 0x7f;
select
{
  Type == 0: ShowGeneralEthernet();
  Type == 1: ShowControl();
  Type == 2: ShowCompressedEthernet();
  Type == 3: ShowCompressedSourceEthernet();
  Type == 4: ShowCompressedDestEthernet();
  default: ShowUnknownBnep();
};
```

```
while (E && ! Bnep_Error)
{
  AddSeparator();
  ShowExtension();
};
```

```
out.DataLength = in.DataLength - DecodedBytes;
GetNBits(in.Data, DecodedBytes * 8, 0);
CopyToOutData( out.DataLength );
}
```

```
BuildCellList()
{
  AddSeparator();
```

```

GetNBits(out.Data, 0, 0);
ShowDataBlock(out.DataLength);
}

```

```
##### 0x00- General Ethernet BNEP type #####
```

```

ShowGeneralEthernet()
{
  AddCell("Type","GenEthrnet", "0x00- General Ethernet BNEP type", Bnep_color);

  if ( IsNotEnoughBnepData(14, 1) ) return 0;

  data1 = GetNBits(in.Data, 8, 4 * 8);
  data2 = NextNBits(2 * 8);
  AddCell("Destination",Format("0x%08X", data1) + Format("%04X", data2),"Destination Ad-
dress",Bnep_color );

  data1 = NextNBits(4 * 8);
  data2 = NextNBits(2 * 8);
  AddCell("Source",Format("0x%08X", data1) + Format("%04X", data2),"Source Ad-
dress",Bnep_color );

  protocol = NextNBits(2 * 8);
  ShowEtherProtocol(protocol);

  DecodedBytes = 15;
}

```

```
##### 0x01- Control BNEP type #####
```

```

const BNEP_control_table = [
  [ 0x07, "Reserved"          , "Reservedfor future use"    , "" ],

```



```

[ 0x00, "COMMAND_NOT_UNDERSTOOD" ,
"BNEP_CONTROL_COMMAND_NOT_UNDERSTOOD" , "ShowCtrlNotUnderstood" ],
[ 0x01, "SETUP_CONN_REQ"      , "BNEP_SETUP_CONNECTION_REQUEST_MSG"  ,
"ShowCtrlSetupReq" ],
[ 0x02, "SETUP_CONN_RES"     , "BNEP_SETUP_CONNECTION_RESPONSE_MSG" ,
"ShowCtrlSetupRes" ],
[ 0x03, "FILTER_NET_TYPE_SET" , "BNEP_FILTER_NET_TYPE_SET_MSG"      , "Show-
FilterNetTypeSet" ],
[ 0x04, "FILTER_NET_TYPE_RES" , "BNEP_FILTER_NET_TYPE_RESPONSE_MSG" ,
"ShowFilterNetTypeRes" ],
[ 0x05, "FILTER_MULTI_ADDR_SET" , "BNEP_FILTER_MULTI_ADDR_SET_MSG"    ,
"ShowFilterMultiAddrSet" ],
[ 0x06, "FILTER_MULTI_ADDR_RES" , "BNEP_FILTER_MULTI_ADDR_RESPONSE_MSG" , "ShowFilterMultiAddrRes" ]
];

```

```
ShowControl()
```

```
{
```

```
AddCell("Type","Control", "0x01- Control BNEP type", Bnep_color);
```

```
if ( IsNotEnoughBnepData(1, 1) ) return 0;
```

```
control_type = NextNBits(8);
```

```
control_row = Find(BNEP_control_table, control_type);
```

```
AddCell("Control", control_row[1], Format("0x%02X - ", control_type) + control_row[2],
Bnep_color);
```

```
DecodedBytes = DecodedBytes + 1;
```

```
Call (control_row[3]);
```

```
}
```

```
ShowCtrlNotUnderstood()
```

```
{
```

```
if ( IsNotEnoughBnepData(1, 1) ) return 0;
```

```
ctrl = NextNBits(8);
```

```
control_row = Find(BNEP_control_table, control_type);
```

```
AddCell("UnknownControl", Format("0x%02X", ctrl), "Unknown Control Type - " +
control_row[2], Bnep_color);
```

```
DecodedBytes = DecodedBytes + 1;
}

ShowCtrlSetupReq()
{
    if ( IsNotEnoughBnepData(1, 1) ) return 0;

    uuid_len = NextNBits(8);
    AddCell("UUID", Format("%d bytes", uuid_len), "length of the SDP service UUIDs",
    Bnep_color);
    DecodedBytes = DecodedBytes + 1;

    if ( IsNotEnoughBnepData(uuid_len, 1) ) return 0;
    i = uuid_len;
    while ( i > 0)
    {
        str = str + Format( "%02X", NextNBits(8) );
        i = i - 1;
    }
    ShowDestUUID(str);
    DecodedBytes = DecodedBytes + uuid_len;

    if ( IsNotEnoughBnepData(uuid_len, 1) ) return 0;

    i = uuid_len;
    str = "";
    while ( i > 0)
    {
        str = str + Format( "%02X", NextNBits(8) );
        i = i - 1;
    }
    ShowSourceUUID(str);
    DecodedBytes = DecodedBytes + uuid_len;
}
}
```

```
const BNEP_resp_msg_table = [  
  [ 0x0005 , "Reserved"      , "Reserved for future use"      ],  
  [ 0x0000 , "Success"      , "Operation Successful"      ],  
  [ 0x0001 , "Bad Dest UUID" , "Operation Failed: Invalid Destination Service UUID" ],  
  [ 0x0002 , "Bad Source UUID" , "Operation Failed: Invalid Source Service UUID" ],  
  [ 0x0003 , "Bad UUID Size" , "Operation Failed: Invalid Service UUID Size" ],  
  [ 0x0004 , "not allowed" , "Operation Failed: Connection not allowed" ]  
];
```

```
ShowCtrlSetupRes()
```

```
{  
  if ( IsNotEnoughBnepData(2, 1) ) return 0;  
  
  data = NextNBits(16);  
  
  row = Find(BNEP_resp_msg_table, data);  
  AddCell("message", row[1], Format("0x%04X - ", data) + row[2], Bnep_color);  
  DecodedBytes = DecodedBytes + 2;  
}
```

```
ShowFilterNetTypeSet()
```

```
{  
  if ( IsNotEnoughBnepData(2, 1) ) return 0;  
  
  length = NextNBits(16);  
  AddCell("list length", Format("%d", length), "Length of the start and end range of network protocol  
types list", Bnep_color);  
  DecodedBytes = DecodedBytes + 2;  
  
  if ( IsNotEnoughBnepData(length, 1) ) return 0;
```

```
decoded = 0;
while ((length - decoded) > 3)
{
    start = NextNBits(16);
    AddCell("Start",Format("0x%04X", start), "Network Protocol Type range Start", Bnep_color);

    end = NextNBits(16);
    AddCell("End",Format("0x%04X", end), "Network Protocol Type range End", Bnep_color);

    decoded = decoded + 4;
    DecodedBytes = DecodedBytes + 4;
}
}
```

```
const BNEP_filter_resp_msg_table = [
    [ 0x0005 , "Reserved"          , "Reserved for future use"          ],
    [ 0x0000 , "Success"           , "Operation Successful"             ],
    [ 0x0001 , "Unsupported"       , "Unsupported Request"             ],
    [ 0x0002 , "Bad range"         , "Operation Failed: Invalid Networking Protocol Type Range" ],
    [ 0x0003 , "Limited Reached"   , "Operation Failed: Maximum Filter Limited Reached" ],
    [ 0x0004 , "Unable due to security" , "Operation Failed: Unable to fulfill request due to security"
]
];
```

```
ShowFilterNetTypeRes()
```

```
{
    if ( IsNotEnoughBnepData(2, 1) ) return 0;

    data = NextNBits(16);

    row = Find(BNEP_filter_resp_msg_table, data);
    AddCell("message", row[1], Format("0x%04X - ", data) + row[2], Bnep_color);
}
```

```
DecodedBytes = DecodedBytes + 2;
}
```

```
ShowFilterMultiAddrSet()
```

```
{
  if ( IsNotEnoughBnepData(2, 1) ) return 0;

  length = NextNBits(16);
  AddCell("list length", Format("%d", length), "Length of the start and end range of multicast IEEE
address range list", Bnep_color);
  DecodedBytes = DecodedBytes + 2;

  if ( IsNotEnoughBnepData(length, 1) ) return 0;

  decoded = 0;
  while ((length - decoded) > 11)
  {
    start = Format( "0x%04X" , NextNBits(16));
    start = start + Format( "%04X" , NextNBits(16));
    start = start + Format( "%04X" , NextNBits(16));

    AddCell("Start", start, "Start of the range of 48 bit Multicast IEEE address not to be filtered",
Bnep_color);

    end = Format( "0x%04X" , NextNBits(16));
    end = end + Format( "%04X" , NextNBits(16));
    end = end + Format( "%04X" , NextNBits(16));
    AddCell("End", end, "Start of the range of 48 bit Multicast IEEE address not to be filtered",
Bnep_color);

    decoded = decoded + 12;
    DecodedBytes = DecodedBytes + 12;
  }
}
```

```
}  
}
```

```
const BNEP_filter_ma_resp_msg_table = [  
  [ 0x0005 , "Reserved"          , "Reserved for future use"          ],  
  [ 0x0000 , "Success"           , "Operation Successful"            ],  
  [ 0x0001 , "Unsupported"       , "Unsupported Request"            ],  
  [ 0x0002 , "Bad Address"       , "Operation Failed: Invalid Multicast Address" ],  
  [ 0x0003 , "Limited Reached"   , "Operation Failed: Maximum Multicast Address Filter Limited Reached" ],  
  [ 0x0004 , "Unable due to security" , "Operation Failed: Unable to fulfill request due to security" ],  
];
```

```
ShowFilterMultiAddrRes()
```

```
{  
  if ( IsNotEnoughBnepData(2, 1) ) return 0;  
  
  data = NextNBits(16);  
  
  row = Find(BNEP_filter_ma_resp_msg_table, data);  
  AddCell("message", row[1], Format("0x%04X - ", data) + row[2], Bnep_color);  
  DecodedBytes = DecodedBytes + 2;  
}
```

```
##### 0x02 - Compressed Ethernet BNEP type #####
```

```
ShowCompressedEthernet()
{
  AddCell("Type","ComprEthernet", "0x02 - Compressed Ethernet BNEP type", Bnep_color);

  if (in.DataLength < 3 )
  {
    ShowBnepError(1);
  }
  else
  {
    protocol = GetNBits(in.Data, 8, 2 * 8);
    ShowEtherProtocol(protocol);

    DecodedBytes = 3;
  }
}
```

```
##### 0x03 - Compressed Ethernet Source Only BNEP type #####
```

```
ShowCompressedSourceEthernet()
{
  AddCell("Type","ComprSource", "0x03 - Compressed Ethernet Source Only BNEP type",
  Bnep_color);

  if (in.DataLength < 9 )
  {
    ShowBnepError(1);
  }
  else
```

```
{
    data1 = GetNBits(in.Data, 8, 4 * 8);
    data2 = NextNBits(2 * 8);
    AddCell("Source",Format("0x%08X", data1) + Format("%04X", data2),"Source Ad-
address",Bnep_color );

    protocol = NextNBits(2 * 8);
    ShowEtherProtocol(protocol);

    DecodedBytes = 9;
}
}
```

##### 0x04 - Compressed Ethernet Destination Only BNEP type  
#####

```
ShowCompressedDestEthernet()
{
    AddCell("Type","ComprDest", "0x04 - Compressed Ethernet Destination Only BNEP type",
Bnep_color);

    if (in.DataLength < 9 )
    {
        ShowBnepError(1);
    }
    else
    {
        data1 = GetNBits(in.Data, 8, 4 * 8);
        data2 = NextNBits(2 * 8);
        AddCell("Destination",Format("0x%08X", data1) + Format("%04X", data2),"Destination Ad-
dress",Bnep_color );

        protocol = NextNBits(2 * 8);
        ShowEtherProtocol(protocol);

        DecodedBytes = 9;
    }
}
```



```
}  
}
```

```
##### Unknown BNEP type #####
```

```
ShowUnknownBnep()
```

```
{  
  AddCell("Type","Unknown", Format("0x%02X - Unknown BNEP type", Type), Bnep_color);  
  
  DecodedBytes = 1;  
}
```

```
##### Extensions #####
```

```
ShowExtension()
```

```
{  
  if ( IsNotEnoughBnepData(1, 2) ) return 0;  
  
  byte = GetNBits(in.Data, DecodedBytes * 8, 8);  
  DecodedBytes = DecodedBytes + 1;  
  
  E = (byte >> 7) & 1;  
  AddCell("E",Format ("%d", E),"Extension flag", Bnep_color);  
  
  ext_type = byte & 0x7f;  
  select  
  {  
    ext_type == 0: ShowExtCtrl();  
    default: ShowExtUnknown(ext_type);  
  };  
}
```

```
ShowExtCtrl()
```

```
{
```

```
  AddCell("ExtType","Control", "0x00 - Extension Control", Bnep_color);
```

```
  if ( IsNotEnoughBnepData(1, 2) ) return 0;
```

```
  ext_length = GetNBits(in.Data, DecodedBytes * 8, 8);
```

```
  DecodedBytes = DecodedBytes + 1;
```

```
  AddCell("ExtLength",Format ("%d", ext_length),"Extension Length", Bnep_color);
```

```
  if ( IsNotEnoughBnepData(ext_length, 2) ) return 0;
```

```
  control_type = NextNBits(8);
```

```
  control_row = Find(BNEP_control_table, control_type);
```

```
  AddCell("Control", control_row[1], Format("0x%02X - ", control_type) + control_row[2],  
  Bnep_color);
```

```
  DecodedBytes = DecodedBytes + 1;
```

```
  Call (control_row[3]);
```

```
}
```

```
ShowExtUnknown(byte)
```

```
{
```

```
  AddCell("ExtType","Unknown", Format("0x%X - Reserved Extension Type value", byte),  
  Bnep_color);
```

```
}
```

```
const BNEP_SDP_UUID_table = [
```

```

[ "x" , "" , "unknown UUID" ],
[ "1115" , "PANU" , "PAN user" ],
[ "1116" , "NAP" , "Network Access Point" ],
[ "1117" , "GN" , "General Networking t" ]
];

```

ShowDestUUID(str)

```

{
  uuid_row = Find(BNEP_SDP_UUID_table, str);
  AddCell("Dest UUID",
    select {
      uuid_row[1] == "" : "0x" + str;
      uuid_row[1] == null: "0x" + str;
      default : uuid_row[1];
    }
    , "Destination SDP service UUID\n" + "0x" + str + " - " + uuid_row[2],
    Bnep_color);
}

```

ShowSourceUUID(str)

```

{
  uuid_row = Find(BNEP_SDP_UUID_table, str);
  AddCell("Source UUID",
    select {
      uuid_row[1] == "" : "0x" + str;
      uuid_row[1] == null: "0x" + str;
      default : uuid_row[1];
    }
    , "Source SDP service UUID\n" + "0x" + str + " - " + uuid_row[2],
    Bnep_color);
}

```

ShowEtherProtocol(protocol)

```

{
  protocol_row = Find( EthernetProtocolsTable, protocol);
  AddCell("NetProtocol",

```

```
select {
    protocol_row[1] == "" : Format ("0x%04X", protocol);
    protocol_row[1] == null: Format ("0x%04X", protocol);
    default      : protocol_row[1];
}
    ,"Networking Protocol Type\n" + Format ("0x%04X - ", protocol) + protocol_row[2],
Bnep_color);

out.EtherProtocol = protocol;
}

#----- Functions for Error handling-----
IsNotEnoughBnepData(need, error_type)
{
    if ((in.DataLength - DecodedBytes) < need )
    {
        ShowBnepError(error_type);
        return 1;
    }
    return 0;
}

const Bnep_Error_Table =
[
    [0 , "No error" , "No Error" ],
    [1 , "No data" , "not enough data for this type of transaction" ],
    [2 , "No data" , "Not enough data for extension option"]
];
```

```
ShowBnepError(error)
{
  if (error != null)
  {
    Bnep_Error = error;
    AddCell("Error", Bnep_Error_Table[error][1], Bnep_Error_Table[error][2] , RED,
    _MONOCOLOR);
  }
}
```

